


# KaSim reference manual v1.06

`KappaLanguage.org`

June 26, 2011




This document is work in progress...

# Contents



# Chapter 1

## Introduction



KaSim-logo.png

## 1.1 Preamble

This manual describes the usage of **KaSim**, the latest implementation of Kappa, one member of the growing family of rule-based languages. Rule-based modelling has attracted recent attention in developing biological models that are concise, comprehensible, and easily extensible. Although this manual contains a self-contained description of Kappa, it is *not* intended as a tutorial on rule-based modeling.<sup>1</sup>

## 1.2 The KaSim engine

**KaSim** is an open source stochastic simulator of rule-based models [?, ?, ?] written in the  $\kappa$ -calculus. Basically **KaSim** takes one or several kappa files as input and generates stochastic trajectories of various observables. **KaSim** implements Danos and Krivine’s simulation algorithm [?] which adapts Gillespie’s algorithm [?, ?] to rule-based models.

A *simulation event* corresponds to the application of a rewriting rule, contained in the kappa file, to the current graph (also called a *mixture*). The rule is selected according to its *activity*, ie the number of instances it has in the current mixture, multiplied by its kinetic rate, and applied to one of its possible instances in the graph. It results in a new graph together with an updated activity for all rules (see Fig. ??). Importantly, the cost of a simulation event is independent of the size of the graph it is applied to [?].

Note that **KaSim** is not equipped with a curve visualization tool. However, data outputs are given in a text format that is readily usable by any standard plotting software such as gnuplot.

## 1.3 Support

- Kappa language tutorials and downloads: <http://kappalanguage.org>
- Bug reports should be posted on github: <https://github.com/jkrivine/KaSim/issues>
- Questions and answers on the kappa-user mailing list: <http://groups.google.com/group/kappa-users>
- Want to contribute to the project? `jkrivine at pps dot jussieu dot fr`

---

<sup>1</sup>For an idea of how Kappa is used in a modeling context, the reader might find the following short note useful Agile modelling of cellular signalling (SOS’08). A longer article, expounding on the causal aspects of modeling, is also available: Rule-based modelling of cellular signalling (CONCUR’07), See also this tutorial: Modelling epigenetic information maintenance: a Kappa tutorial (CAV’09).




Figure 1.1: The event loop





## Chapter 2

# Installation

### 2.1 Obtaining the sources

To obtain KaSim you can either use pre-compiled binaries available on [KappaLanguage.org](http://KappaLanguage.org) or compile the sources for your architecture. To do so, download the source code from <https://github.com/jkrivine/KaSim> and make sure you have a recent ocaml compiler installed. From a terminal window type `ocamlopt.opt -v`. If nothing appears then you need to install Ocaml Native compiler that can be downloaded from <http://caml.inria.fr/download.en.html>.

### 2.2 Compilation

Once Ocaml is safely installed, untar KaSim archive and compile following these few steps:

```
$ tar xzvf kasim.tar.gz -d Kappa
$ cd Kappa
$ make
```

At the end of these steps you should see, in the `Kappa` directory, an executable file name `KaSim`. In order to check the compilation went fine, simply type `.\KaSim --version`. If the ocaml native compiler `ocamlopt.opt` is not in the path of your system, you may set the variable `OCAMLBINPATH` to point to the location of the compiler by editing the corresponding line in the Makefile.



# Chapter 3

## The command line

### 3.1 General usage

From a terminal window, KaSim can be invoked by typing

```
$ KaSim -i file_1 ... -i file_n [option]
```

where `file_i` are the input kappa files containing the rules, initial conditions and observables, see Chapter ?? below. Tables ?? and ?? summarize all the options that can be given to the simulator. Basically, one specifies an upper bound either in bio time (arbitrary time unit), or in number of events. Note that bio-time is computed using Gillespie's formula for time advance (see Fig. ??) and should not be confused with CPU-time (it's not even proportional). In doubt, we recommend using a bound in number of events since the cost of one event application is bounded (in CPU time) by a constant, so the simulation time of  $n$  events is roughly  $k$  times faster than a simulation of  $k * n$  events.

### 3.2 Main options

Table ?? summarizes the main options that are accessible through the command line. Note that the option `-p` specifies the number of points that one wishes to have in the final plot. The interval at which these points will be taken is then computed using the simulation limit defined by the user using option `-t` or `-e`. For instance requiring a simulation of 100 points during 10 time units will result in a simulation where observables are recorded every 0.1 time units, while requiring 100 points for 1000 events will result in observable being written every 10 events (be careful that the event density, ie the number of events per time unit, might vary during a simulation, and thus the two sampling methods can lead to very different repartitions of the moments at which observables are recorded even if they sepecify the same number of observations).

The results of the options `-im` and `-flux` is explained Chapter ??.

Table 3.1: Command line: main options

Argument	Description
<b>-e</b> $e_{max}$	Terminates simulation after $e_{max} \geq 0$ events
<b>-t</b> $t_{max}$	Terminates simulation after $t_{max} \geq 0.0$ time units
<b>-p</b> $n$	Produces a data file (default: <b>data.out</b> ) with $n \geq 0$ data points
<b>-o</b> $file$	Set the name of data file to $file$
<b>-d</b> $dir$	Output any produced file to the directory $dir$
<b>-im</b> $file$	Produces a dot format file of the decorated influence map of the model
<b>-flux</b> $file$	Produces a dot format file of the activation/inhibition flux measured during the simulation

### 3.3 Advanced options

Table ?? summarizes the advanced options that are accessible through the command line.

Table 3.2: Command line: advanced options

Argument	Description
<b>--seed</b> $n$	Seeds the pseudo-random number generator $n > 0$
<b>--implicit-signature</b>	Automatically deduce agent signatures (see Chapter ??)
<b>--dot-format</b>	Produces dot format files for snapshots (instead of kappa files)

### Example

The command

```
$ KaSim -i model.ka -e 1000000 -p 1000 -o model.out
```

will generate a file **model.out** containing the trajectories of the observables defined in the kappa file **model.ka**. The file **model.out** will contain 1000 data points (ie in this case, a measure will be taken every 1000 events). The command

```
$ KaSim -i init.ka -i rules.ka -i obs.ka -i mod.ka -t 1.5 -p 1000
```

will generate a file **data.out** (default name) containing 1000 data points of a simulation of 1.5 seconds (arbitrary time units) of the model. Note that the input kappa file is split in

4 files containing, for instance, the initial conditions, `init.ka`, the rule set, `rules.ka`, the observables, `obs.ka`, and the perturbations, `pert.ka` (refer to Chapter ?? for details). The order in which the files are given does not matter.

### 3.3.1 The activation and flux maps

The command line options `-im` and `-flux` allows modelers to generate additional information about the statics and the dynamics of the model that is being considered.



## Chapter 4

# The kappa file

### 4.1 General remarks

The *Kappa File* (KF) is the formal representation of your model. We use KF to denote the union of the files that are given as input to **KaSim** (argument `-i`). Each line of the KF is interpreted by **KaSim** as a *declaration*. If the line is ended by the escape character `'\'` the continuation of the declaration is parsed onto the next line. Declarations can be of 4 types: *signatures* (Sec. ??), *rules* (Sec. ??), *variables* (Sec. ??), *initial conditions* (Sec. ??) and *perturbations* (Sec. ??). The KF's structure is quite flexible and can be divided in any number of sub-files in which the order of declarations does not matter (to the exception of variable declarations, see Section ?? for details). Comments can be used by inserting the marker `#` that tells **KaSim** to ignore the rest of the line.

### 4.2 Agent signature

*Agent signatures* constitute a form of typing information about the agents that are used in the model. It contains information about the name and number of interaction sites the agent has, and about their possible internal states. A signature is declared in the KF by the following line:

```
%agent: signature_expression
```

according to the grammar given Table ??. For instance the line:

```
%agent: A(x,y~u~p,z~0~1~2) # Signature of agent A
```

will declare an agent **A** with 3 (*interaction*) sites **x**, **y** and **z** with the site **y** possessing two *internal states* **u** and **p** (for instance for the unphosphorylated and phosphorylated forms of **y**) and the site **z** having possibly 3 states respectively 0, 1 and 2. Note that internal states values are treated as untyped symbols by **KaSim**, so choosing a character or an integer as internal state is purely matter of convention.

<i>signature_expression</i>	::=	<b>Id</b> ( <i>sig</i> )
<i>sig</i>	::=	<b>Id</b> <i>internal_state_list</i> , <i>sig</i>   $\varepsilon$
<i>internal_state_list</i>	::=	<b>~Id</b> <i>internal_state_list</i>   $\varepsilon$

Table 4.1: Agent signature expression: terminal symbol are denoted in (blue) typed font. Symbol **Id** can be any string generated by regular expression  $[a-zA-Z0-9][a-zA-Z0-9\_ -]^*$ . Terminal symbol  $\varepsilon$  stands for the empty symbol.

### 4.3 Rules

Once agents are declared, one may add to the KF the rules that describe their dynamics through time. Roughly a Kappa rule looks like

‘my rule’ *kappa\_expression*  $\rightarrow$  *kappa\_expression* @  $\gamma$

where ‘my rule’ can be any name that will refer to the subsequent rule that can be decomposed into a *left hand side* (LHS) and a *right hand side* (RHS) kappa expressions together with a *kinetic rate* *k*. Kappa expressions are generated by the grammar given Table ??.

<i>kappa_expression</i>	::=	<i>agent_expression</i> , <i>kappa_expression</i>   $\varepsilon$
<i>agent_expression</i>	::=	<b>Id</b> ( <i>interface</i> )
<i>interface</i>	::=	$\varepsilon$   <b>Id</b> <i>internal_state</i> <i>link_state</i>
<i>internal_state</i>	::=	$\varepsilon$   <b>~Id</b>
<i>link_state</i>	::=	$\varepsilon$   <b>!n</b>   <b>!_</b>   <b>?</b>

Table 4.2: Kappa expressions: In addition to the conventions of Table ??, symbol **n** denotes any positive integer.

#### 4.3.1 A simple rule

With the signature of **A** defined in the previous section, the line

‘A dimerization’ **A**(**x**),**A**(**y~p**)  $\rightarrow$  **A**(**x!1**),**A**(**y~p!1**) @  $\gamma$

denotes a dimerization rule between two instances of agent **A** provided the second is phosphorylated (say that is here the meaning of **p**) on site **y**. Note that the bond between both **As** is denoted by the identifier **!1** which uses an arbitrary integer (**!0** would denote the same bond. In Kappa, a bond may connect exactly 2 sites so any occurrence of a bond identifier **!n** has to be paired with exactly one other sibling in the expression. Note also the fact that site **z** of **A** is not mentioned in the expression which means that it has no influence on the triggering of this rule. This is the *don’t care don’t write convention* (DCDW) that plays a key role in resisting combinatorial explosion when writing models.



### 4.3.2 Adding and deleting agents

Sticking with **A**'s signature, the rule

$$\text{'budding A'} \quad A(z) \rightarrow A(z!1), A(x!1) @ \gamma$$

indicates that an agent **A** free on site **z**, no matter what its internal state is, may beget a new copy of **A** bound to it via sites **x**. Note that in the RHS, agent **A**'s interface is not completely described. Following the DCDW convention, **KaSim** will then assume that the sites that are not mentioned are created in the *default state*, ie they appear free of any bond and their internal state (if any) is the first of the list shown in the signature (here state **u** for **y** and **0** for **z**).

Importantly, **KaSim** respects the *longest prefix convention* to determine which agent in the RHS stems from an agent in the LHS. In a word, from a rule of the form  $a_1, \dots, a_n \rightarrow b_1, \dots, b_k$ , with  $a_i$ s and  $b_j$ s being agents, one computes the biggest indices  $i \leq n$  such that the agents  $a_1, \dots, a_i$  are pairwise consistent with  $b_1, \dots, b_i$ , ie the  $a_j$ s and  $b_j$ s have the same name and the same number of sites. In which case we say that for all  $j \leq i$ ,  $a_j$  is *preserved* by the transition and for all  $j > i$ ,  $a_j$  is *deleted* by the transition and  $b_j$  is *created* by the transition. This convention allows us to write a deletion rule as:

$$\text{'deleting A'} \quad A(x!1), A(z!1) \rightarrow A(x) @ \gamma$$

which will remove the **A** agent in the mixture that will match the second occurrence of **A** in this rule.

### 4.3.3 Side effects

It may happen that the application of a rule has some *side effects* on agents that are not mentioned explicitly in the rule. Consider for instance the previous rule:

$$\text{'deleting A'} \quad A(x!1), A(z!1) \rightarrow A(x) @ \gamma$$

The **A** in the graph that is matched to the second occurrence of **A** in the LHS will be deleted by the rule. As a consequence all its sites will disappear together with the bonds that were pointing to them. For instance, when applied to the graph

$$G = A(x!1, y\tilde{p}, z\tilde{2}), A(x!2, y\tilde{u}, z\tilde{0!1}), C(t!2)$$

the above rule will result in a new graph  $G' = A(x!1, y\tilde{p}, z\tilde{2}), C(t)$  where the site **t** of **C** is now free as side effect.

*Wildcard* symbols for link state ? (for bound or not), !\_ (for bound to someone), may also induce side effects when they are not preserved in the RHS of a rule, as in

$$\text{'Disconnect A'} \quad A(x!_) \rightarrow A(x) @ \gamma$$

or

$$\text{'Force bind A'} \quad A(x?) \rightarrow A(x!1), C(t!1) @ \gamma$$

Both these rule will cause **KaSim** to raise a warning at rule compile time.

### 4.3.4 Kinetic rates

Rules are equipped with a *kinetic rate*, denoted by  $\gamma$  above. For **KaSim**, this number should correspond to the *stochastic rate constant* at which the corresponding reaction should be applied. We refer the reader to <http://www.KappaLanguage.org> for a more complete tutorial on kinetic rates, but basically  $\gamma$  corresponds to the *deterministic rate constant*  $k$  of the reaction corrected by the volume  $V$  in which the model is considered.

We often want to express  $\gamma$  in "per molecule" rather than "per mol". This only changes the numerical value, not the dimension (both are numbers of molecules):

$$\gamma = \frac{k}{(\mathcal{A} V)^{(a-1)}} \quad [molecule^{-1} s^{-1}], \quad (4.1)$$

where  $\mathcal{A} = 6.022 \cdot 10^{23}$  is Avogadro's number,  $k$  in molar units  $M^{-1} s^{-1}$  and  $a \geq 0$  is the arity of the rule (ie 2 for a bimolecular rule)

- Mammalian cell:  $V = 2.25 \cdot 10^{-12} l$  ( $1l = 10^{-3} m^3$ ), and  $\mathcal{A}V = 1.35 \cdot 10^{12}$ .

A concentration of  $1M$  in a mammalian cell volume corresponds to  $1.35 \cdot 10^{12}$  molecules;  $1nM \approx 1350$  molecules per cell.

- Yeast cell (haploid):  $V = 4 \cdot 10^{-14} l$ , and  $\mathcal{A}V = 2.4 \cdot 10^{10}$ .

A concentration of  $1M$  in a yeast cell volume corresponds to  $2.4 \cdot 10^{10}$  molecules;  $1nM \approx 24$  molecules per cell. The volume is doubled in a diploid cell.

- E.coli cell:  $V = 10^{-15} l$ , and  $\mathcal{A}V = 10^8$ .

A concentration of  $1M$  in a yeast cell volume corresponds to  $10^8$  molecules;  $10nM \approx 1$  molecule per cell.

The following table lists a few ballpark figures for deterministic rate constants and their stochastic counterparts in a mammalian cell volume:

process	$k$	$\gamma$	stoch. dimension
general binding	$10^7 - 10^9$	$10^{-5} - 10^{-3}$	$molecule^{-1} s^{-1}$
general unbinding	$10^{-3} - 10^{-1}$	$10^{-3} - 10^{-1}$	$s^{-1}$
dephosphorylation	1	1	$s^{-1}$
phosphorylation	0.1	0.1	$s^{-1}$
receptor dimerization	$2 \cdot 10^6$	$1.6 \cdot 10^{-6}$	$molecule^{-1} s^{-1}$
receptor dissociation	$1.6 \cdot 10^{-1}$	$1.6 \cdot 10^{-1}$	$s^{-1}$

## 4.4 Variables

In the KF it is also possible to declare *variables* by a line of the form

```
%var: 'var_name' (variable_expression or kappa_expression)
```

where `var_name` can be any string and *variable\_expression* is any algebraic expression on variable names (other than `var_name`) using predefined operators summarized in Table ??.

Symbol	Interpretation
[E]	the number of simulation events since the beginning of the simulation
[T]	the bio-time of the simulation
'v'	the value of variable 'v'
[f]	the intuitive mathematical function or constant associated to $f \in \{\log, \sin, \cos, \tan, \text{sqrt}, \text{pi}\}$
[inf]	symbol for $\infty$
[mod]	the <i>modulo</i> operator
[exp]	the exponentiation operation $x \mapsto e^x$
[int]	the integer part $x \in \mathbb{R} \mapsto  x  \in \mathbb{N}^+$
+, -, *, /, ^	the corresponding mathematical operators

Table 4.3: Symbol usable in variable expressions.

For instance the declarations

```
%var: 'homodimer' A(x!1),A(x!1)
```

```
%var: 'aa' 'homodimer'/2
```

define 2 variables, the first one tracking the number of embeddings of  $A(x!1), A(x!1)$  in the graph over time, while the second divides this value by 2: the number of automorphisms in  $A(x!1), A(x!1)$ . The declaration of a variable acts as a binder in the rest of the KF. Hence variables that are used in the expression of another variable must be declared beforehand.

It is also possible to use algebraic expressions as kinetic rates as in

```
%var: 'k_on' 1.0E-6 # per molecule per second
```

```
'ab' A(x),A(x) -> A(x!1),A(x!1) @ 'k_on'/2
```

KaSim may output values of variables in the data file (see option `-p` in Chapter ??) using plot declaration:

```
%plot: 'var_name'
```

One may use the shortcut:

```
%obs: 'var_name' variable_expression
```

to declare a variable and at the same time require it to be outputted in the data file.

## 4.5 Initial conditions

The initial mixture to which rules in the KF will be applied are declared as

```
%init: n kappa_expression
```

where  $n > 0$  can be any integer. This will add to the initial state of the model  $n$  copies of the graph described by the kappa expression. Again the DCDW convention allows us not to write the complete interface of added agents (the remaining sites will be completed according to the agent's signature). For instance:

```
%init: 1000 (A,A(y~p))
```

will add 1000 instances of **A** in its default state  $A(x,y\sim u,z\sim 0)$  and 1000 instances of **A** in state  $A(x,y\sim p,z\sim 0)$ . Note that **A** is equivalent to writing **A()**. As any other declaration, **%init** can be used multiple times, and agents will add up to the initial state.

## Chapter 5

# A simple model

We describe below the content of a simple Kappa model and give examples of some typical run<sup>1</sup>.

### 5.1 ABC.ka

```
1. ##### Signatures
2. %agent: A(x,c) # Declaration of agent A
3. %agent: B(x) # Declaration of B
4. %agent: C(x1~u~p,x2~u~p) # Declaration of C with 2 modifiable sites
5. ##### Rules
6. 'a.b' A(x),B(x) -> A(x!1),B(x!1) @ 'on_rate' #A binds B
7. 'a..b' A(x!1),B(x!1) -> A(x),B(x) @ 'off_rate' #AB dissociation
8. 'ab.c' A(x!_,c),C(x1~u) ->A(x!_,c!2),C(x1~u!2) @ 'on_rate' #AB binds C
9. 'mod x1' C(x1~u!1),A(c!1) ->C(x1~p),A(c) @ 'mod_rate' #AB modifies x1
10. 'a.c' A(x,c),C(x1~p,x2~u) -> A(x,c!1),C(x1~p,x2~u!1) @ 'on_rate' #A binds C on x2
11. 'mod x2' A(x,c!1),C(x1~p,x2~u!1) -> A(x,c),C(x1~p,x2~p) @ 'mod_rate' #A modifies x2
12. ##### Variables
13. %var: 'on_rate' 1.0E-4 # per molecule per second
14. %var: 'off_rate' 0.1 # per second
15. %var: 'mod_rate' 1 # per second
16. %obs: 'AB' A(x!x.B)
17. %obs: 'Cuu' C(x1~u,x2~u)
18. %obs: 'Cpu' C(x1~p,x2~u)
19. %obs: 'Cpp' C(x1~p,x2~p)
```

---

<sup>1</sup>The corresponding kappa file is included in the distribution of KaSim, in the directory `models/`

```

20. ##### Initial conditions
21. %init: 1000 A,B
22. %init: 10000 C

```

Line 1-4 of this KF contains signature declarations. Agents of type **C** have 2 sites **x1** and **x2** whose internal state may be **u**(nphosphorylated) or **p**(hosphorylated). Recall that the default state of these sites is **u** (the first one). Line 8, rule '**ab.c**' binds an **A** connected to someone on site **x** (link type **!\_**) to a **C**. Note that the only rule that binds an agent to **x** of **A** is '**a.b**' at line 6. Hence the use of **!\_** is a commodity and the rule could be replaced by

```
'alt_ab.c' A(x!1,c),B(x!1),C(x1~u) → ...
```

There are two main points to notice about this model: **A** can modify both sites of **C** once it is bound to them. However, only an **A** bound to a **B** can connect on **x1** and only a free **A** can connect on **x2**. Note also that **x2** is available for connection only when **x1** is already modified.

## 5.2 Some runs

We try first a coarse simulation of 100,000 events (10 times the number of agents in the initial system).

```
$ KaSim -i ABC.ka -e 100000 -p 1000 -o abc.out
```

Plotting the content of the **abc.out** file one notices that nothing of significant interest happen to the observables after 250s. So we can now specify a meaningful time limit by running

```
$ KaSim -i ABC.ka -e 100000 -t 250 -o abc.out
```

which produces the data points whose rendering is given Fig. ?? . We will use this model as a running example for the next chapter, in order to illustrate various advanced concepts.




Figure 5.1: Simulation of the ABC model: population of unmodified **Cs** (observable **Cuu** in red) drops rapidly and is replaced, in a first step by simply modified **Cs** (observable **Cpu** in blue) which are in turn replaced by doubly modified **Cs** (observable **Cpp** in red). Note that the population of **AB** complexes (observable **AB** in black) stabilizes slightly below 400 individuals after about 20s.





## Chapter 6

# Advanced concepts

### 6.1 Perturbation language

#### 6.1.1 Syntax

It is possible to use variables of the model as precondition for triggering a *perturbation* of the simulation. For instance one may want to interrupt the simulation whenever a certain amount of observable has been reached, introduce a signal at some time interval etc. To do so, one declares a perturbation with a line:

```
%mod: boolean_expression do effect end_expression
```

where *boolean\_expression* and *effect* are defined by the grammar given Table ???. Basically,

```
boolean_expression ::= algebraic_expression rel algebraic_expression
                    | (boolean_expression || boolean_expression)
                    | (boolean_expression && boolean_expression)
                    | [not] boolean_expression
                    | [true] | [false]
```

```
effect ::= $ADD algebraic_expression agent_expression
        | $DEL algebraic_expression agent_expression
        | $SNAPSHOT | $STOP
        | 'rule_name' := algebraic_expression
```

```
end_expression ::= ε | until boolean_expression
```

Table 6.1: Perturbation expressions. The operator `rel` can be any binary relation in  $\{<, =, >\}$

the *precondition* determines when the perturbation will be triggered, for instance a user writes

```
%mod: ([T]>10) && ('v1'/'v2') > 1 do ...
```

to indicate she wishes to trigger a perturbation whenever the simulation time has passed 10 time units and the ratio of variables *v1* over *v2* is above 1.

The *effect* of a perturbation can be of 4 kinds: adding (\$ADD instruction) or deleting (\$DEL instruction) agents from the mixture, recording the current state of the mixture (\$SNAPSHOT instruction), interrupting the simulation and saving the last mixture (\$STOP instruction) or setting the kinetic rate of a particular rule to a given value.

Continuing with the ABC model, the perturbation effect:

```
$ADD n C(x1~p)
```

will add  $n \geq 0$  instances of *C* with *x1* already in state *p* (and the rest of its interface in the default state as specified line 4 of ABC.ka). Also the perturbation effect:

```
$DEL [inf] B(x!_)
```

will remove *all* Bs connected to someone from the mixture. Eventually, the effect:

```
'rule_name' := [inf]
```

will set the kinetic rate of a declared rule named '*rule\_name*' to  $\infty$ .

The *end\_expression* serves to specify whether the perturbation should be applied only once ("one shot" perturbation) or at every future events until a condition is reached (persistent perturbation). For instance:

```
%mod: ([modulo] [T] 100)=0 do $DEL [inf] C until [T]>1000
```

will delete every *C* from the mixture every 100 t.u until the simulation clock has passed 1000 t.u. If no *until* keyword is positioned, the perturbation will be discarded after its first application.

### 6.1.2 Example 1: introducing new agents during the simulation

There are various ways one can use perturbations to study more deeply a given kappa model. A basic illustration is the use of a simple perturbation to let a system equilibrate before starting a real simulation. For instance, as can be seen from the curve given Fig. ??, the number of AB complexes is arbitrarily set to 0 in the initial state (all As are disconnected from Bs in the initial mixture. In order to avoid this, one can modify the kappa file the following way: we set the initial concentration of *C* to 0 by deleting line 22. Now we introduce Cs after 25 t.u by perturbation:

```
%mod: [T]>25 do $ADD 10000 C
```

The modified kappa file is available in the source repository, in the `model/` directory (file `abc-pert.ka`). Running again a simulation (a bit longer) by entering in the command line:

```
$ KaSim -i ABC-pert.ka -e 100000 -t 300 -o abc2.out
```

one obtains the curve given Fig. ??




Figure 6.1: Simulation of the ABC model with a perturbation: for  $t < 25$ s, only ‘a.b’ and ‘a..b’ rules may apply. This enables the concentration of ‘AB’ complexes to go to steady state, before introducing fresh Cs at  $t = 25$ s.

### 6.1.3 Example 2: Defining a new initial state

In the previous example, we let the system evolve for some time without its main reactant C in order to let other reactants go to a less arbitrary initial state. One may object that this way of proceeding is CPU-time consuming if one has to do this at each simulation. An alternative is to use the `$SNAPSHOT` primitive that allows a user to export a snapshot of the mixture at a given time point as a new (piece of) kappa file. For instance, the declaration:

```
%mod: [T]>25 do $SNAPSHOT
```

will ask KaSim to export the mixture after 25 time units of simulation. The exported file will be named `snap_n.ka` where  $n$  is the event number at which the snapshot was taken. Note that if the name already exists a counter will be appended at the end of the file to prevent overwriting. We can now replace the perturbation of the previous example by the above one and let the system run for 26 time units. The simulation will produce a readable file that one may now use as new initial condition that can be used as input of KaSim using `-i` argument.

Note that snapshots can be performed multiple times, for instance every 1000 events, using the declaration:

```
%mod: ([mod] [E] 1000)=0 do $SNAPSHOT until [false]
```

which results in KaSim producing a snapshot every 1000 events until the simulation ends. The perturbation `$STOP` will terminate the simulation whenever its precondition is satisfied and produce a snapshot of the last mixture.

## 6.2 Link type

In standard kappa, in order to require a site to be bound for an interaction to occur, one may use the *semi-link* construct `!_` which does not specify who the partner of the bond is. For instance in the variable:

```
%var: 'ab' A(x!_),B(y!_)
```

will count the number of As and Bs connected to someone, including the limit case `A(x!1),B(y!1)`. It is sometimes convenient to specify the *type* of the semi-link, in order to restrict the choice of the binding partner. For instance the variable:

```
%var: 'ab' A(x!y.B),B(y!x.A)
```

will count the number of As whose site `x` is connected to a site `y` of B, plus the number of Bs whose site `y` is connected to a site `x` of A. Note that this still includes the case `A(x!1),B(y!1)`.

## 6.3 Implicit signature

KaSim permits users in a hurry to avoid writing agent signatures explicitly using the option `--implicit-signature` of the command line. The signature is then deduced using information gathered in the KF. Note that it is not recommended to use the DCDW convention for introduced agents in conjunction with the `--implicit-signature` option unless the default state of all sites is mentioned in the `%init` declarations or in the rules that create agents.

## 6.4 Influence map

The influence map of a model is an object that may help modelers checking the consistency of the rule set they use. It is generated by the command:

```
KaSim -i abc.ka -im map.dot -e 0
```

This will generate the so called *influence map* of the `abc.ka` file (the option `-e 0` specifies that no simulation should be run). The influence map is statically computed and does not depend on kinetic rates nor initial conditions. It describes how rule may potentially influence each other during a simulation. **KaSim** will produce a dot format file containing the influence relation over all rules and observables of the model. The produced graph visualized using a circular rendering<sup>1</sup> is given Figure ?? . Observables are represented as circular nodes<sup>2</sup> and rules as rectangular nodes. Edges are decorated with the list of embeddings (separated by a semi-colon) allowing rules's right hand sides to be mapped to left hand sides.

More formally, consider the rules  $r : L \rightarrow R$  and  $s : L' \rightarrow R'$ . One wishes to know whether it is possible that a given instance of rule  $r$  is a graph  $G$  creates a new instance of rule  $s$ . To do so, **KaSim** will try to generate a  $\kappa$ -term  $T$  containing a match for  $R$  and  $L'$  that overlap on some sites that are modified by  $r$  (see Figure ?? for illustration).

## 6.5 Flux map


The *flux map* is a powerful observation that tracks on the fly the influence rule application have on each other. Contrary to the influence map, it is dynamically generated and tracks effective impacts (positive or negative) a every rule application. The flux map is obtained by the command:

```
KaSim -i abc.ka -im map.dot -t n
```

The resulting *flux map* is a graph where a positive edge between rules  $r$  and  $s$  (in green) indicates an overall positive contribution of  $r$  over  $s$ . Said otherwise, the sum of  $r$  applications increased the activity of  $s$ . Conversely, a negative edge (in red) will indicate that  $r$  had an overall negative impact on the activity of  $s$ . Note that visualization software permitting the importance of the flux between two rules can be observed by looking at the thickness of the edge between the rules. An example of flux map is given Fig. ?? . The generated dot file contains information about the importance of the negative or positive influence of a rule over another. This contribution is represented by the thickness of the edge that is proportional to the flux of activity that passed from one rule to the other. Whenever the flux is too thin to be represented by a plain edge, it is shown dotted. Note that flux may vary during time, therefore the time or event limit of the simulation is of importance and will likely change the aspect of the produced map.

<sup>1</sup>One may use for instance the `circo` program that is part of the *graphviz* suite.

<sup>2</sup>Observable can be viewed as identity rules for computing influence.



abc-im.png

Figure 6.2: The influence map of the `abc.ka` model defined Chapter ?? . Note that observable  $C(x1 \sim u?, x2 \sim u?)$  is not activated by any other rule. This implies that its number of occurrences does not change during a simulation. Edge labels denote embeddings with the convention that  $[i \rightarrow j]$  denotes the embedding of agent number  $i$  of the origin's rhs, into agent  $j$  of the target's lhs.




Figure 6.3: Computation of the influence of the top rule on the rule below: the right hand side of the first rules embeds in a common term with the left hand side of the second rule. It results that the first rule has a positive influence on the second.




Figure 6.4: Flux map of the `abc.ka` model, taken from  $t=0$  to  $t=20$  time units. Note that the rules `ab.c` and `mod x1` are the biggest contributors to activity update in the simulated interval.